

```
> restart;
```

```
[restart erases all variables
```

```
> with(simplex):
```

```
Warning, the protected names maximize and minimize have been redefined and unprotected
```

```
[Notation:
```

```
mi = barrels of crude oil processed by method i
```

```
g = grade
```

```
G = Gasoline
```

```
H = heating oil
```

```
k = cracked
```

```
> eq_g6:= g6k+g6G+g6H = .2*m1+.3*m2+.4*m3;
```

$$eq\_g6 := g6k + g6G + g6H = 0.2 m1 + 0.3 m2 + 0.4 m3$$

```
[This expresses the fact that grade 6 produced = grade 6 used
```

```
> eq_g8:= g8k+g8G+g8H = .3*m1+.4*m2+.4*m3+g6k;
```

$$eq\_g8 := g8k + g8G + g8H = 0.3 m1 + 0.4 m2 + 0.4 m3 + g6k$$

```
[This expresses the fact that grade 8 produced = grade 8 used
```

```
> eq_g10:= g10G+g10H = .5*m1+.3*m2+.2*m3+g8k;
```

$$eq\_g10 := g10G + g10H = 0.5 m1 + 0.3 m2 + 0.2 m3 + g8k$$

```
> eq_G:= G = g6G+g8G+g10G;
```

$$eq\_G := G = g6G + g8G + g10G$$

```
> eq_G9:= g6G*6+g8G*8+g10G*10 >= 9*G;
```

$$eq\_G9 := 9 G \leq 6 g6G + 8 g8G + 10 g10G$$

```
> eq_Gmax:= G <= 2000;
```

$$eq\_Gmax := G \leq 2000$$

```
> eq_H:= H = g6H + g8H + g10H;
```

$$eq\_H := H = g6H + g8H + g10H$$

```
> eq_H7:= g6H*6 + g8H*8 + g10H*10 >= 7*H;
```

$$eq\_H7 := 7 H \leq 6 g6H + 8 g8H + 10 g10H$$

```
> eq_Hmax:= H <= 600;
```

$$eq\_Hmax := H \leq 600$$

```
> constraints:= {eq_g6, eq_g8, eq_g10, eq_G, eq_G9, eq_Gmax, eq_H, eq_H7, eq_Hmax};
```

```
constraints := {g8k + g8G + g8H = 0.3 m1 + 0.4 m2 + 0.4 m3 + g6k,  
g10G + g10H = 0.5 m1 + 0.3 m2 + 0.2 m3 + g8k, G = g6G + g8G + g10G,  
9 G ≤ 6 g6G + 8 g8G + 10 g10G, G ≤ 2000, H = g6H + g8H + g10H, 7 H ≤ 6 g6H + 8 g8H + 10 g10H,  
H ≤ 600, g6k + g6G + g6H = 0.2 m1 + 0.3 m2 + 0.4 m3}
```

```
> revenue:= 11 * G + 6 * H;
```

```
revenue := 11 G + 6 H
```

```
> cost:= 3.4*m1 + 3*m2 + 2.6 *m3 + 1*g6k + 1.5*g8k;
```

```
cost := 3.4 m1 + 3 m2 + 2.6 m3 + g6k + 1.5 g8k
```

```
> profit:= revenue-cost;
```

```
profit := 11 G + 6 H - 3.4 m1 - 3 m2 - 2.6 m3 - g6k - 1.5 g8k
```

```
> sols:=maximize(profit,constraints,NONNEGATIVE);
```

```
sols := {g8k = 480.0000000, G = 2000.0000000, H = 600.0000000, g10H = 0., g6G = 0., m1 = 0.,  
g6H = 300.0000000, g6k = 740.0000000, g10G = 1000.0000000, g8G = 1000.0000000,  
m3 = 2600.0000000, g8H = 300.0000000, m2 = 0.}
```

```
> subs(sols,profit);
```

```
17380.00000
```

```
>
```